8/31/99

Notes for beam bending problem 9/2

1. Set up the cantilever beam as beam theory

Define the load on the upper surface and write in terms of total force, F, figured positive if pointing up

```
STUDENT > q:=p*w*(x-L):
          int(q,x=0..L):
          p:=solve("=F,p):
          q;
```

$$-2\,\frac{F\,(x-L)}{L^2}$$

Integrate to get shear flow and moment

```
STUDENT > V:=V0-int(q,x):
          Mb:=M0-int(V,x):
```

Set shear flow and moment equal to zero at the right hand end and solve for V0 and M0

```
STUDENT > x:=L:V0:=solve(V,V0):M0:=solve(Mb,M0):x:='x':
```

```
STUDENT > V0,M0;
```

$$F,\frac{1}{3}F\,L$$

And factor the shear flow and moment for convenience

```
STUDENT > V:=factor(V);Mb:=factor(Mb);
```

$$V := \frac{F\,(L-x)^2}{L^2}$$

$$Mb := \frac{1}{3}\,\frac{F\,(L-x)^3}{L^2}$$

2. Put in the approximate stresses that give the resultant shear flow and moments, working from the class notes

(Solve for the principal strains in terms of the stresses first.)

```
STUDENT > e1:=sx-2*mu*ex-lambda*(ex+ey+ez):
          e2:=sy-2*mu*ey-lambda*(ex+ey+ez):
          e3:=sz-2*mu*ez-lambda*(ex+ey+ez):
          solve({e1,e2,e3},{ex,ey,ez}):
STUDENT > assign("):
          ex:=collect(ex,[sx,sy,sz]):
          ey:=collect(ey,[sx,sy,sz]):
          ez:=collect(ez,[sx,sy,sz]):
          exs:=0:eys:=0:ezs:=0:
```

```
for i from 1 to 3 do
exs:=exs+factor(op(i,ex)):
eys:=eys+factor(op(i,ey)):
ezs:=ezs+factor(op(i,ez)):od:
exs;eys;ezs;
```

$$\frac{(\mu+\lambda)\,sx}{\mu\,(2\,\mu+3\,\lambda)}-\frac{1}{2}\frac{\lambda\,sy}{\mu\,(2\,\mu+3\,\lambda)}-\frac{1}{2}\frac{sz\,\lambda}{\mu\,(2\,\mu+3\,\lambda)}$$

$$-\frac{1}{2}\frac{\lambda\,sx}{\mu\,(2\,\mu+3\,\lambda)}+\frac{(\mu+\lambda)\,sy}{\mu\,(2\,\mu+3\,\lambda)}-\frac{1}{2}\frac{sz\,\lambda}{\mu\,(2\,\mu+3\,\lambda)}$$

$$-\frac{1}{2}\frac{\lambda\,sx}{\mu\,(2\,\mu+3\,\lambda)}-\frac{1}{2}\frac{\lambda\,sy}{\mu\,(2\,\mu+3\,\lambda)}+\frac{(\mu+\lambda)\,sz}{\mu\,(2\,\mu+3\,\lambda)}$$

```
STUDENT > ex:=exs:ey:=eys:ez:=ezs:
```

Now I can plug in the model stresses from the class notes

```
STUDENT > sx:=-12*Mb*y/w/h^3;
          sy:=12*q*(1/8*h^2*y-1/6*y^3+1/24*h^3)/w/h^3;
          sz:=0;
```

$$sx := -4\,\frac{F\,(L-x)^3\,y}{L^2\,w\,h^3}$$

$$sy := -24\,\frac{F\,(x-L)\left(\frac{1}{8}h^2\,y-\frac{1}{6}y^3+\frac{1}{24}h^3\right)}{L^2\,w\,h^3}$$

$$sz := 0$$

And work on the strains

```
STUDENT > collect(ex,y);
```

ex = du/dx, so I can integrate this, remembering that u(0) = 0.

```
STUDENT > int(ex,x=0..x1):u:=subs(x1=x,");
STUDENT > factor(op(1,u))+factor(op(2,u));
```

$$\frac{1}{4}F\,(4\,y\,\mu\,L^4-16\,y\,\mu\,L^3\,x+24\,y\,\mu\,L^2\,x^2-16\,y\,\mu\,L\,x^3+4\,y\,\mu\,x^4+4\,y\,\lambda\,L^4-16\,y\,\lambda\,L^3\,x$$
$$+24\,y\,\lambda\,L^2\,x^2-16\,y\,\lambda\,L\,x^3+4\,y\,\lambda\,x^4+3\,\lambda\,x^2\,h^2\,y-6\,\lambda\,x\,h^2\,y\,L-4\,\lambda\,x^2\,y^3+8\,\lambda\,x\,y^3\,L+\lambda\,x^2\,h^3$$
$$-2\,\lambda\,x\,h^3\,L)\,/\,(L^2\,w\,h^3\,\mu\,(2\,\mu+3\,\lambda))-\frac{(\mu+\lambda)\,F\,L^2\,y}{w\,h^3\,\mu\,(2\,\mu+3\,\lambda)}$$

```
STUDENT > u:=":
```

Similarly, I can get v by integrating ey, but I need to add in the displacement of the neutral axis, v0(x)

```
STUDENT > ey;
STUDENT > int(ey,y=0..y1):v:=subs(y1=y,")+v0(x);
STUDENT > factor(op(1,"))+factor(op(2,"));
```

$$\frac{1}{2} F\, y\, (L-x)\,(2\lambda y L^2 - 4\lambda y x L + 3\mu h^2 y - 2\mu y^3 + 2\mu h^3 + 3\lambda h^2 y - 2\lambda y^3 + 2\lambda h^3 + 2\lambda y x^2) \Big/ (\mu\,(2\mu+3\lambda)\, L^2\, w\, h^3) + \mathrm{v0}(x)$$

```
STUDENT > v:=":
```

Now the shear is given by dudy + dvdx, so I can calculate this from the candidate displacements

```
STUDENT > gxy:=diff(u,y)+diff(v,x);
```

$$gxy := \frac{1}{4} F\,(4\mu L^4 - 16\mu L^3 x + 24\mu L^2 x^2 - 16\mu L x^3 + 4\mu x^4 + 4\lambda L^4 - 16\lambda L^3 x + 24\lambda L^2 x^2 - 16\lambda L x^3 + 4\lambda x^4 + 3\lambda x^2 h^2 - 6\lambda x h^2 L - 12\lambda x^2 y^2 + 24\lambda x y^2 L) \Big/ (L^2 w h^3 \mu\,(2\mu+3\lambda)) - \frac{(\mu+\lambda) F L^2}{w h^3 \mu\,(2\mu+3\lambda)} - \frac{1}{2}\frac{F\, y\,(2\lambda y L^2 - 4\lambda y x L + 3\mu h^2 y - 2\mu y^3 + 2\mu h^3 + 3\lambda h^2 y - 2\lambda y^3 + 2\lambda h^3 + 2\lambda y x^2)}{\mu\,(2\mu+3\lambda)\, L^2 w h^3} + \frac{1}{2}\frac{F\, y\,(L-x)\,(-4\lambda y L + 4\lambda y x)}{\mu\,(2\mu+3\lambda)\, L^2 w h^3} + \left(\frac{\partial}{\partial x}\mathrm{v0}(x)\right)$$

The shear stress is known from out guess from beam theory

```
STUDENT > txy:=6*V/w/h^3*((h/2)^2-y^2);
```

$$txy := 6\,\frac{F\,(L-x)^2\left(\frac{1}{4}h^2 - y^2\right)}{L^2\, w\, h^3}$$

and we take the difference between the shear stress divided by mu and the shear as calculated. (This would to be zero were the solution from beam theory exact).

```
STUDENT > txy/mu-gxy;
```

We need to find dv0dx to make this zero (if that's possible)

```
STUDENT > solve(",diff(v0(x),x)):
STUDENT > temp:=collect(",y):
STUDENT > test:=0:
          for i from 1 to nops(temp) do
          test:=test+factor(op(i,temp)) od:
          dv0dx:=test;
```

We know that h and y are of the same order, so let's gather the system up in terms of them

```
STUDENT > collect(",[y,h],distributed);
```

We can write y in terms of a dimensionless parameter eta (-1 ≤ eta ≤ 1) and h

```
STUDENT > y:=eta*h/2:collect(dv0dx,h);
```

And we see there are term that go like 1/h^3, 1/h and h. h has dimensions. so that's not really quite enough to allow me to say that one term is bigger than another. h will be small compared to L, however, and we can write

```
STUDENT > h:=epsilon*L:dv0dx:=collect(dv0dx,epsilon);
```

And the we can also scale L, introducing xi ($0 \le xi \le 1$)

```
STUDENT > x:=L*xi:""":
```

So here we have the nice scaled version of dv0dx

```
STUDENT > factor(op(1,dv0dx))+factor(op(2,dv0dx))+factor(op(3,dv0dx));
```

$$-\frac{1}{16}\frac{F(\mu+\lambda)\eta(-6\eta-8+\eta^3)\varepsilon}{L\,w\,\mu(2\mu+3\lambda)}-\frac{3}{4}F(4\eta^2\xi^2\mu-8\eta^2\lambda\xi-8\eta^2\xi\mu+4\eta^2\lambda\xi^2+4\eta^2\mu+5\eta^2\lambda-4\xi^2\mu-4\mu+8\xi\mu+10\lambda\xi-6\lambda-5\lambda\xi^2)/(L\,w\,\mu(2\mu+3\lambda)\varepsilon)-\frac{F\xi(\xi-2)(\xi^2-2\xi+2)(\mu+\lambda)}{L\,w\,\mu(2\mu+3\lambda)\varepsilon^3}$$

The largest term is the third one, and it is independent of eta (hence y), so we see that for h/L sufficiently small, the beam theory solution is a good approximation to the (unknown) elasticity solution. We integrate the leading term to get the deflection, and add in a constant to make the deflection at the origin zero

```
STUDENT > deflection:=L*int(op(3,"),xi)+Cv;
```

```
STUDENT > xi:=0:Cv:=solve(deflection,Cv):xi:='xi':
          deflection:=factor(deflection);
```

$$deflection := -\frac{1}{5}\frac{F(\mu+\lambda)\xi^2(\xi^3-5\xi^2+10\xi-10)}{w\,\mu(2\mu+3\lambda)\varepsilon^3}$$

The variation is in the numerator, so we can set up for a dimensionless plot

```
STUDENT > num:=numer(deflection);
STUDENT > factor("/F/(mu+lambda));
STUDENT > plotdef:=":
          plot(plotdef,xi=0..1);
```



the maximum deflection is at xi = 1 (x = L)

```
STUDENT > xi:=1:maxdef:=factor(deflection);xi:='xi':
```

$$maxdef := \frac{4}{5}\frac{F(\mu+\lambda)}{w\,\mu\,(2\,\mu+3\,\lambda)\,\varepsilon^3}$$

And that agrees with the result in Crandall & Dahl (solution to problem 8.14)